
Coq/SSReflectの論理

1 Coqの論理

CoqはCalculus of Inductive Constructions (CIC)という型理論に基いている．型理論は型付
きλ計算という計算体系と直観主義論理の証明論の間の同型 (カリー・ハワード対応)に基いた論
理体系である．CICは特に表現力が高い (一階述語論理や高階論理を含んでいる)が，まずその命
題論理の部分から見ていくことにする

1.1 直観主義の命題論理

論理式 論理式は以下の結合子から定義される．

P,Q ::= True | False 定数
| A 論理変数
| P → Q 含意
| P ∧Q 論理積
| P ∨Q 論理和

否定はないが，便宜のために ¬P = P → False とおく．

導出規則 自然演繹体系では真の論理式は以下の規則より導出される．
∆ を論理式の集合とする．Trueは常に∆に含まれる．

公理　 ∆ ⊢ P (P は∆に含まれる)

→導入 ∆, P ⊢ Q

∆ ⊢ P → Q

→除去 ∆ ⊢ P ∆ ⊢ P → Q

∆ ⊢ Q

矛盾　
∆ ⊢ False

∆ ⊢ P

∧導入 ∆ ⊢ P ∆ ⊢ Q

∆ ⊢ P ∧Q

∧除去 ∆ ⊢ P ∧Q

∆ ⊢ P

∆ ⊢ P ∧Q

∆ ⊢ Q

∨導入 ∆ ⊢ P

∆ ⊢ P ∨Q

∆ ⊢ Q

∆ ⊢ P ∨Q

∨除去 ∆ ⊢ P ∨Q ∆, P ⊢ R ∆, Q ⊢ R

∆ ⊢ R

この導出規則から背理法や排中律が導けない．もしも古典論理の体系が欲しければ，矛盾の代
わりに以下の規則のどちらかを使う．

背理法
∆,¬P ⊢ False

∆ ⊢ P
¬¬除去 ∆ ⊢ ¬¬P → P

恒真式 命題論理の恒真式は Trueだけを仮定して導出できる式である．
例えば，P → P ∧ P や P → (P → Q) → Q は恒真式である．それぞれの導出を以下に示す．

P ∧Q ⊢ P ∧Q

P ∧Q ⊢ Q

P ∧Q ⊢ P ∧Q

P ∧Q ⊢ P
P ∧Q ⊢ Q ∧ P

⊢ P ∧Q → Q ∧ P

(公理)

(∧除去)

(∧導入)

(→導入)

P, P → Q ⊢ P P, P → Q ⊢ P → Q

P,P → Q ⊢ Q

P ⊢ (P → Q) → Q

⊢ P → (P → Q) → Q

(公理)

(→除去)

(→導入)

(→導入)

1.2 Coqでの証明

変数宣言 まずは，準備として論理変数の宣言を行う．Sectionというコマンドを使うと，局所
的な論理変数が宣言できるようになる．宣言自体は Variablesコマンドを使う．そして，宣言範
囲が終ると Endコマンドでセクションを閉じる．Coqの出力をイタリック体で表示している．

Section Koushin.

Variables P Q : Prop.
P is assumed
Q is assumed

変数を宣言するときにその型も書かなければならないが，論理式の型は Propになる．

命題と証明
まず，二つ目の恒真式を証明してみよう．定理を宣言すると証明モードに移る．

Theorem modus_ponens : P -> (P -> Q) -> Q. (* 名前を付けなければならない *)
1 subgoal (* 証明の状況が表示される *)
P, Q : Prop
============================
P -> (P -> Q) -> Q

線の上は∆の中身，下は結論だが，今は∆に変数の宣言しかない．P とQが命題であるという
宣言であり，P やQが成り立つという仮定ではない．→導入にあたるCoqのタクティック (証明
命令)は introsである．→除去は apply．公理は assumption．

Proof.
intros p pq. (* 仮定に名前を付ける (導入) *)

1 subgoal
p : P
pq : P -> Q
============================
Q

apply pq. (* 目標を仮定 pqの帰結とみなす (除去) *)
1 subgoal
p : P
pq : P -> Q
============================
P

assumption. (* (公理) *)
Proof completed.

Qed.
modus_ponens is defined

実際の証明をもう一度みよう．

Theorem modus_ponens : P -> (P -> Q) -> Q.
Proof.
intros p pq.
apply pq.
assumption.

Qed.

証明が長い訳ではないが，少し違和感がある．直感的に二つ目の前提を一つ目に適用すれば結
論が出るはずなのに，逆にそれを結論に「適用」している．証明を結論から作っていくので，考
え方が逆になってしまい，証明が長くなると追いにくい．

2 SSReflectでの証明

上の自然演繹体系はCoqの中の論理によく合っているが，証明の構築には不便である．証明の
構築は，証明したい命題と仮定を下に書き，証明木を延して行く形で行うが，自然演繹の場合で
は，規則は結論 (右側)に対するものばかりで，その結論を変形させる「逆算」で行うことになる．
通常の数学の証明は仮定から性質を導く「前進」方向で行うことが多い．Gentzenのシークエン
ト計算が結論を扱う「右規則」とは別に仮定を扱う「左規則」も持っている．ここでは，そうい
う右規則を自然演繹に追加する形で SSReflectの証明方法を反映した論理体系を作る．

2.1 SSReflectの論理

新しい判定の形は
∆;P ⊢ Q

∆には名前付きの仮定が入っている．P は焦点と言い，左規則の対象となる. 焦点は空白でもい
い．Qは結論といい，右規則の対象である．

Focus
∆;P ⊢ Q

∆;⊢ P → Q

Unfocus
∆;⊢ P → Q

∆;P ⊢ Q

Intro
∆, h : P ;⊢ Q

∆;P ⊢ Q

Revert
∆;P ⊢ Q

∆, h : P ;⊢ Q

Apply-R
∆;⊢ P1 . . . ∆;⊢ Pn

∆;P1 → . . . → Pn → Q ⊢ Q

Apply-L1
∆;⊢ P → Q ∆;Q ⊢ R

∆;P ⊢ R

Apply-L2
∆;⊢ P ∆;Q ⊢ R

∆;P → Q ⊢ R

公理 ∆;P ⊢ P

矛盾 ∆;False ⊢ P

カット
∆;⊢ P ∆;P ⊢ Q

∆;⊢ Q

∧-R ∆;⊢ P ∆;⊢ Q

∆;⊢ P ∧Q

∧-L ∆;P ⊢ Q → R

∆;P ∧Q ⊢ R

∨-R ∆;⊢ P

∆;⊢ P ∨Q

∆;⊢ Q

∆;⊢ P ∨Q

∨-L ∆;P ⊢ R ∆;Q ⊢ R

∆;P ∨Q ⊢ R

規則は下から上へと読む．Focusは結論の前提を焦点に移す．Unfocusは逆に焦点された命題
を結論の前提に戻す．SSReflectではこの二つの規則は暗に使われ，焦点と結論の前提が区別
されない．Introは焦点に名前を付けて仮定に移す．Revertは仮定を焦点に戻す．基本的にはこ
の 4つの規則が仮定を動かしているいるだけだ．Apply-Rは焦点を使って結論を証明するが，そ
の際焦点の全ての前提を証明する必要が新たに生じる．Apply-L1は補題 P → Qを使って焦点
を変える．Apply-L2は逆に補題 P を使って焦点の前提をみたす．この 3つの規則を右側のカッ
トで代用することができるので，論理的には必要ではないが，証明を自然に進めるのに使う．右
側の規則は導入が自然演繹と変らないが，右規則やカットが焦点を利用している．特に ∧-Rでは
焦点に P しか入らないので，Q が前提に戻される．

2.2 SSReflectのタクティック

SSReflectの論理は多くの規則をもっているが，Coqと違い各規則が独立したタクティック
に対応している訳ではない．ほとんどのものが「move+修飾子」という形で書ける．
もう一回 modus ponensを証明する．

Reset modus_ponens. (* module_ponens 以降の定義を忘れる *)
Require Import ssreflect. (* ssreflect のタクティック言語を使う *)

Theorem modus_ponens : P -> (P -> Q) -> Q.
Proof.
move=> p. (* Intro *)

1 subgoal
p : P
============================
(P -> Q) -> Q

move/(_ p). (* Apply-L2 *)
1 subgoal
p : P
============================
Q -> Q

done. (* 公理や矛盾 *)
No more subgoals. (* 証明完了 *)
Restart. (* 証明をやりなおす *)
by move => p /(_ p). (* 全てを一行にまとめる *)

Qed.

move の => と / は一緒にまとめられ，さらに by を他のタクティックの前に書くと，後に done

を加えたとくと同じ意味になる．
もう一つの恒真も試してみよう．

Theorem and_comm : P /\ Q -> Q /\ P.
Proof.
move=> [] p q. (* ∧-L, Intro, Intro *)
p : P
q : Q
============================
Q /\ P

split. (* ∧-R *)
2 subgoals

P, Q : Prop
p : P
q : Q
============================
Q

subgoal 2 is:
P

done. done.
No more subgoals.

Restart.
by move=> [p q]; split. (* 名前を中に書く，タクティックを「;」でつなぐ *)

Qed.

恒真色々 証明状態の表示が作戦を読みにくくするので，これ以降は省くことにする．自分でCoq

の中で実行して，確認して下さい．

Theorem or_comm : P \/ Q -> Q \/ P.
Proof.

move=> [p|q]. (* (∨-L; 証明木が分岐するとき，縦棒「|」で分ける *)
by right. (* ∨-R2 *)

by left. (* ∨-R1 *)
Qed.

Theorem DeMorgan : ~ (P \/ Q) -> ~ P /\ ~ Q.
Proof.
rewrite /not. (* 定義を展開する *)
move=> npq.
split=> [p|q]. (* 全てのタクティックで「=>」が使える *)
apply: npq. (* (Apply-R) *)
by left.

apply: npq.
by right.

Qed.
End Koushin. (* Section を閉じる *)

論理和に関するDe Morganの法則が証明できた．しかし，双対的な法則 (¬(P ∧Q) ⊃ ¬P ∨¬Q)

は直観主義論理ではなりたたない．二重否定の除去か排中律を仮定する必要がある．それらの同
値性を証明した上でその定理を証明する．

Section Classic.
Definition Classic := forall P : Prop, ~ ~ P -> P. (* ¬¬除去の定義 *)
Definition EM := forall P : Prop, P \/ ~ P. (* 排中律の定義 *)

Lemma Classic_is_EM : Classic <-> EM. (* ¬¬除去と排中律が同値 *)
Proof.
rewrite /Classic /EM. (* 定義の展開 *)
split => [classic | em] P. (* A <-> B := A -> B /\ B -> A *)
+ apply: (classic) => nEM. (* classic を仮定から消さずに焦点におく *)
have p : P. (* (カット) *)
apply: classic => np.
apply: nEM. by right.

apply: nEM. by left.
+ by move: (em P) => []. (* Pについての排中律で場合分けをする *)

Qed.

Definitionは項や命題を定義する．ここでは二重否定除去と排中律の言明を定義するのに使う．
forall P : Propはこの言明が任意の命題に対して使えることを表している．
証明の中の「+」(または「-」や「*」)は場合分けの構造を分かりやすくすつために使う．一時

的に他のゴールが見えなくなる．

Variables P Q : Prop.

Theorem DeMorgan’ : Classic -> ~ (P /\ Q) -> ~ P \/ ~ Q.
Proof.
move=> /Classic_is_EM em npq. (* Apply-L1, Intro, Intro *)
move: (em P) => [p|np]. (* 排中律で場合分け *)
+ move: (em Q) => [q|nq].

- elim: npq. (* 矛盾 *)
by split.

- by right.
+ by left.

Qed.
End Koushin. (* Section を閉じる *)
Check DeMorgan’. (* 命題の言明を確認する *)
DeMorgan’

: forall P Q : Prop, Classic -> ~ (P /\ Q) -> ~ P \/ ~ Q

Sectionから出ると，使われた命題変数が各定理の言明に自動で追加される．

論理規則と tacticの対応

論理規則 作戦 論理規則 作戦

Intro move => h 公理 done, by

Revert move: h 矛盾 elim, suff: False

Apply-R apply ∧-R split

Apply-L1 move/e ∧-L move => []

Apply-L2 move/( e) ∨-R left, right

カット have, suff, move: (e) ∨-L move => [|]

各タクティックの詳しい説明は次回 (述語論理)のときに述べる．

練習問題 2.1 以下の定理をCoqで証明せよ．

Section Coq1.
Variables P Q R : Prop.
Theorem imp_trans : (P -> Q) -> (Q -> R) -> P -> R.
Theorem not_false : ~False.
Theorem double_neg : P -> ~~P.
Theorem contraposition : (P -> Q) -> ~Q -> ~P.
Theorem and_assoc : P /\ (Q /\ R) -> (P /\ Q) /\ R.
Theorem and_distr : P /\ (Q \/ R) -> (P /\ Q) \/ (P /\ R).
Theorem absurd : P -> ~P -> Q.
Definition DM_rev := forall P Q, ~ (~P /\ ~Q) -> P \/ Q.
Theorem DM_rev_is_EM : DM_rev <-> EM.

End Coq1.

6


